


IL PRIMO SETTIMANALE DI SOFTWARE SU CARTA

L. 1.000

PER IL TUO PERSONAL COMPUTER

Una pubblicazione della J.soft editrice



# PAPER soft

19

Anno 1 - N° 19 - 2 novembre 1984



**Bandiere  
Totocalcio**



**Roader  
Gestione files dati su cassetta**



**Guerre stellari  
La foresta stregata**



**Diagrammi a torta su  
printer-plotter 1520**



**Sing-Sing  
Castle dungeon**

TI99IUC.it

Editrice

**J.soft**

Con la collaborazione del Gruppo Editoriale Jackson



**TI-99/4A**

**10**

**Roader**

di K. Smith trad. e adatt. di E. Re Garbagnati

extended basic

**TI-99/4A**

**12**

**Gestione files dati su cassetta**

di M. Cristuib Grizzi

J. soft s.r.l.

**DIREZIONE, REDAZIONE,  
AMMINISTRAZIONE**

Via Rosellini, 12  
20124 MILANO  
Tel. (02) 68.88.228

**DIRETTORE RESPONSABILE:**

Pietro Dell'Orco

**COORDINAMENTO TECNICO:**

Riccardo Paolillo

**REDAZIONE:**

Lucio Bragagnolo  
Mauro Cristuib Grizzi

**GRAFICA E IMPAGINAZIONE:**

Margherita La Noce  
Ivana Rossi  
Raffaella Toffolatti

**FOTOCOMPOSIZIONE:**

d&b Via Vignola, 5  
Tel. 02/59.85.08  
20133 MILANO

**CONTABILITÀ:**

Giulia Pedrazzini  
Flavia Bonaiti

**AUTORIZZAZIONE ALLA  
PUBBLICAZIONE:**

Tribunale di Milano n° 200  
del 14.04.1984

**STAMPA:**

Elcograf  
Beverate (CO)

**PUBBLICITÀ**

Concessionario per l'Italia e l'Estero Reina s.r.l.  
Via Washington, 50  
20046 MILANO  
Tel. (02) 49.88.066 (5 linee R.A.)  
Tlx. 316213 REINA I

Concessionario esclusivo per la  
DIFFUSIONE in Italia e Estero:  
SODIP - Via Zuretti, 25  
20125 MILANO

Spedizione in abbonamento  
postale Gruppo II/70  
Prezzo della rivista L. 1.000  
Numeri arretrati L. 2.000

© TUTTI I DIRITTI DI  
RIPRODUZIONE O TRADUZIONE  
DEGLI ARTICOLI PUBBLICATI  
SONO RISERVATI

# Roader

*I motori stanno rombando. Voi state attendendo il segnale di partenza per dimostrare che Niki Lauda, in confronto a voi, non è altro che un pivellino! Finalmente scatta il verde... si parte!!!*

*Una curva, un'altra e... CRASH! Vi siete inchiodati contro il guard-rail!*

*Bene, tutto questo diverrà realtà, e senza alcun rischio per la vostra incolumità personale, soltanto digitando il semplice listato di questo programmino Basic per il vostro TI-99/4a.*

*Avrete a disposizione 4 opzioni per la scelta del tipo di percorso e delle difficoltà nella guida del vostro mini-bolide sulla lunga e ventosa*

*strada generata dal computer.*

*Il migliore risulterà chi avrà saputo totalizzare il massimo punteggio dopo aver percorso il maggior numero possibile di chilometri.*

*Per pilotare il bolide non dovrete fare altro che impugnare il Joystick, o pigiare i due tasti recanti i simboli "maggiore di" e "minore di", ed incrociare le dita...*

*Dopo che vi sarete "impastati" (prima o poi capiterà, siatene cert!) comparirà il punteggio totalizzato e la richiesta di ulteriori istruzioni per procedere o terminare.*

*Cosa altro aspettate per battere il listato? Forza!!*

```

100 REM *****
110 REM * R O A D E R *
120 REM *****
130 REM
140 REM TI 99/4A BASIC
150 F=12
160 GOTO 250
170 FOR VOL=1 TO 30 STEP 10
180 CALL SOUND(-1000,-7,VOL)
190 CALL SCREEN(INT(VOL/2.5)+1)
DOWNLOADED FROM WWW.TI99IUC.IT
200 F=19-F
210 CALL COLOR(9,F,1)
220 NEXT VOL
230 CALL COLOR(9,7,1)
240 RETURN
250 CALL CHAR(99,"223E2A08082A3E
2A")
260 CALL CHAR(100,"447C54100A2E3
F7E")
270 CALL CHAR(101,"00080C1E1E3E3
F7E")
280 CALL CHAR(104,"00181818183C3
C00")
290 CALL CHAR(120,"FFFFFFFFFFFF
FFF")
300 CALL CHAR(128,"00FF00FF00FF0
0FF")
310 CALL COLOR(9,5,1)
320 CALL COLOR(10,16,1)
330 CALL COLOR(11,9,1)
340 CALL COLOR(12,2,1)

```

```

350 CALL COLOR(13,9,2)
360 CALL CLEAR
370 CALL SCREEN(15)
380 PRINT
390 PRINT TAB(8);"c R O A D E R
c"
400 FOR T=1 TO 6
410 PRINT
420 NEXT T
430 PRINT "GUIDA USANDO I TASTI
< E > "
440 PRINT
450 PRINT TAB(6);"O USANDO IL JO
YSTICK #1"
460 FOR T=1 TO 5
470 PRINT
480 NEXT T
490 FOR I=110 TO 120 STEP .5
500 CALL SOUND(-150,I,2)
510 NEXT I
520 FOR I=120 TO 110 STEP -.3
530 CALL SOUND(-150,I,2)
540 NEXT I
550 FOR T=110 TO 120 STEP .8
560 CALL SOUND(-75,T,2)
570 NEXT T
580 GOSUB 170
590 CALL SCREEN(15)
600 CALL COLOR(9,5,1)
610 FOR T=1 TO 250
620 NEXT T
630 CALL CLEAR

```

```

640 CALL COLOR(9,8,2)
650 OSKILL=1
660 PRINT TAB(2);"SCEGLI IL TIPO
DI PERCORSO:"
670 FOR T=1 TO 3
680 PRINT
690 NEXT T
700 PRINT TAB(4);"PREMI : "
710 PRINT
720 PRINT
730 PRINT TAB(8);"1 PER NEO/PATE
NTATO"
740 PRINT
750 PRINT TAB(8);"2 PER TASSISTA
"
760 PRINT
770 PRINT TAB(8);"3 PER RALLYSTA
"
780 PRINT
790 PRINT TAB(8);"4 PER PILOTA D
I F.1"
800 FOR D=1 TO 3
810 PRINT
820 NEXT D
830 CALL KEY(1,K,S)
840 RANDOMIZE
850 IF S=0 THEN 830
860 IF K<>18 THEN 890
870 SKILL=OSKILL
880 GOTO 910
890 SKILL=- (K=19) - (K=7) * 2 - (K=8) *
3 - (K=9) * 4
900 IF SKILL=0 THEN 830
910 OSKILL=SKILL
920 PRINT
930 PRINT TAB(8);"SI PARTE!!!!!"

940 PRINT
950 PRINT
960 FOR T=1 TO 400
970 NEXT T
980 CALL CLEAR
990 CALL SCREEN(3)
1000 I=.1
1010 IF (SKILL<>2) * (SKILL<>4) THE
N 1030
1020 I=.2
1030 N=24
1040 J=0
1050 OLDN=24
1060 B$="xxxxxxxx"
1070 FOR C=1 TO 4
1080 PRINT TAB(18);"h";B$;"h"
1090 NEXT C
1100 FOR Q=1 TO 4
1110 FOR C=9 TO 6 STEP -1
1120 IF C<>9 THEN 1140
1130 B$="xxxxxxxx"
1140 FOR A=0 TO 6.25 STEP 1

```

```

1150 Y=COS(A)
1160 J=J+1
1170 PRINT TAB(8*Y+10);"h";B$;"h
"
1180 IF (RND>.5) + (SKILL=1) + (SKIL
L=2) THEN 1210
1190 IF J<25 THEN 1210
1200 CALL HCHAR(23,28*RND+2,128)
DOWNLOADED FROM WWW.TI9IUC.IT
1210 CALL GCHAR(20,N,G)
1220 CALL HCHAR(19,OLDN,120)
1230 IF (G=104) + (G=128) + (G=32) TH
EN 1440
1240 CALL HCHAR(20,N,99)
1250 OLDN=N
1260 CALL KEY(0,K,S)
1270 IF S<>0 THEN 1290
1280 CALL JOYST(1,XR,YR)
1290 N=N + (K=44) - (K=46) + XR/4
1300 NEXT A
1310 B$=SEG$(B$,1,C-2)
1320 FOR D=110 TO 129-C STEP .5

1330 CALL SOUND(-150,D,2)
1340 NEXT D
1350 NEXT C
1360 NEXT Q
1370 CALL CLEAR
1380 CALL SCREEN(11)
1390 PRINT TAB(5);"CE L'HAI FATT
A,GILLES!!"
1400 FOR T=1 TO 10
1410 PRINT
1420 NEXT T
1430 GOTO 1530
1440 CALL HCHAR(20-1,N,101)
1450 CALL HCHAR(20,N,100)
1460 GOSUB 170
1470 CALL SCREEN(3)
1480 FOR T=1 TO 500
1490 NEXT T
1500 CALL CLEAR
1510 CALL COLOR(9,8,2)
1520 GOTO 1560
1530 FOR I=1 TO 500
1540 NEXT I
1550 CALL CLEAR
1560 CALL SCREEN(15)
1570 PRINT TAB(6);"HAI FATTO";J*
10*SKILL;"P.TI!"
1580 FOR T=1 TO 5
1590 PRINT
1600 NEXT T
1610 PRINT TAB(8);"? GIOCHI ANCO
RA ?"
1620 PRINT
1630 PRINT
1640 PRINT TAB(4);"PREMI: <TASTO
ROSSO> O <S>"

```

```

1650 PRINT TAB(8);"- PER UN PERC
ORSO IDENTICO."
1660 PRINT
1670 PRINT TAB(4);"<Ç> - PER CAM
BIARE LIVELLO"
1680 PRINT
1690 PRINT TAB(4);"<E> - PER TER
MINARE"


```

```

1700 CALL KEY(1,K,S)
1710 IF S=0 THEN 1700
1720 IF (K=18)+(K=2) THEN 980
1730 IF (K<>5)*(K<>14) THEN 1700
DOWNLOADED FROM WWW.TI99IUC.IT
1740 IF K=14 THEN 630
1750 END

```

## Gestione files dati su cassetta



Extended basic

Questo programma è stato scritto per il TI 99/4a fornito del modulo SSS Extended Basic. Permette di usare l'unità a cassette CSI come memoria di massa per un archivio di dati, con numerose opzioni per una completa gestione dei files sequenziali.

Una volta mandato in esecuzione, il programma vi chiederà se volete usare dati precedentemente salvati su nastro; se rispondete affermativamente, verranno visualizzati i parametri che il computer si aspetta di trovare nel file dati ed il tempo di attesa necessario per leggere l'intero file. Comparirà quindi il menu principale, comprendente sette opzioni di scelta: aggiunta (registrazione) di un nuovo record, ricerca di un dato record nel file, display di tutti i records del file con automatica eliminazione di eventuali records vuoti, cancellazione di un record, correzione e modifica di un record, ordinamento alfabetico del file, ed infine uscita - fine lavoro.

Se di sceglie l'opzione di aggiunta record, comparirà un sub-menu con quattro ulteriori opzioni:

- 1 inserzione da tastiera
- 2 inserzione da tastiera in posizione x
- 3 inserzione sequenziale da un file
- 4 inserzione random da un file

La prima opzione rappresenta una normale inserzione di un record in coda ai precedenti; con la seconda è possibile inserire il record in una posizione qualsiasi nel file: se ad esempio il file

è già composto da 100 records, con questa opzione potremo inserire il record in posizione 5 ottenendo uno "scroll" di tutti i records seguenti di una posizione, mentre con l'opzione 1 ci verrà chiesto il 101esimo record. Con l'opzione 3 si può aggiungere al file in memoria un numero qualunque di records prelevati in modo sequenziale da un altro file su nastro, mentre con l'opzione 4 si può ottenere un'inserzione pseudo-random di un record da un altro file su nastro.

Questa opzione richiede particolare attenzione nell'uso, in quanto occorre posizionare il nastro nel punto in cui si prevede ci sia il record che vogliamo inserire, premere ENTER all'apparire dei messaggi 'REWIND CASSETTE TAPE THEN PRESS ENTER' e 'PRESS CASSETTE PLAY THEN PRESS ENTER', e premere il tasto PLAY sul registratore solo quando le sopraccitate scritte hanno avuto un piccolo scroll verso l'alto sul vostro monitor. A questo punto verrà inserito in memoria il record desiderato.

La ricerca di un dato record può essere effettuata sia in base al numero che in base ai primi caratteri del record stesso, mentre correzioni e modifiche sono effettuate linea per linea su tutti i 140 caratteri standard del record. Quando avete finito di creare il vostro archivio o di operare su di esso, sinceratevi che l'ultimo record contenga solo i caratteri "000", che rappresentano un ulteriore controllo di fine del file (benché il programma provveda automati-

camente a registrare in testa al file il numero di records contenuti).

Premendo il tasto 7 del menu principale vi verrà domandato se intendete salvare il file dati ap-

pena creato e, alla risposta affermativa, verranno visualizzati i parametri di struttura del file con il tempo di attesa necessario alla registrazione.

```
110 GOTO 150
120 ON ERROR 130
130 CALL ERR(I,K,W,F):: CALL SOUND(100,500,0)
140 PRINT "WARNING!ERROR in line";F: "error code:";I: "error type";K: "RIPETERE LA PROCEDURA"
  : FOR W=1 TO 1000 :: NEXT W :: RETURN 290
150 CALL SCREEN(2):: FOR I=0 TO 14 :: CALL COLOR(I,13,2):: NEXT I
160 CALL CLEAR :: PRINT "GESTIONE FILES CASSETTE": "*****"
  : "*****": "BY MCG SOFTWARE 1983": : :
170 PRINT "compatibili file con meno di 500 records": : :
180 PRINT : : : : : : PRINT "vuoi usare dati da memoria di massa?(S/N) " :: GOSUB 1320
190 IF KEY=ASC("N")OR KEY=ASC("n")THEN 290
200 CALL CLEAR :: PRINT "PARAMETRI DI DEFAULT": : "org:sequencial": "file type:internal": "record len:fixed 140": "open mode:input"
210 PRINT "file #:1": "lista stampa:alfan.4 input" :: OPEN #1:"CS1",INTERNAL,FIXED 140,INPUT
220 ON ERROR 130 :: INPUT #1:A$ : : K=VAL(A$):: CALL CLEAR :: DISPLAY AT(2,3):"ATTESA": : DISPLAY AT(6,1)SIZE(4):"min." :: DISPLAY AT(6,5)SIZE(4):INT(K*90/60)/10
230 N=1 :: FOR I=1 TO K :: ON ERROR 130 :: INPUT #1:A$,B$,C$,D$: : S$=A$&" "&B$&" "&C$&" "&D$: : MI=INT((K-I)*90/60)/10 :: DISPLAY AT(6,5)SIZE(4):MI :: IF S$="" THEN 270
250 F=3 :: GOSUB 780
260 IF S$="000" THEN 280
270 NEXT I
280 CLOSE #1
290 CALL CLEAR :: MA=500
300 DISPLAY AT(6,3):"1 per registrazione" :: DISPLAY AT(8,3):"2
```

```
per ricerca" :: DISPLAY AT(10,3):"3 per file display/elim." :: DISPLAY AT(11,9):"record vuoti"
310 DISPLAY AT(13,3):"4 per eliminare" :: DISPLAY AT(15,3):"5 per correggere" :: DISPLAY AT(17,3):"6 per ord. alpha" :: DISPLAY AT(19,3):"7 per uscire" :: GOSUB 1320
320 FOR I=49 TO 55 :: IF KEY=I THEN 340
330 NEXT I :: GOTO 32767
340 KEY=KEY-48 :: ON KEY GOSUB 360,580,670,690,1030,1090,770
350 GOTO 290
360 REM AGGIUNTA RECORD
370 PRINT "record";N+1: "1-inserzione da tastiera": "2-da tastiera in posiz. x": "3-inserz.sequenz. da un file": "4-inserz. random da un file"
380 GOSUB 1320
390 IF KEY=50 THEN GOSUB 1140 :: RETURN
400 IF KEY=51 THEN 430
410 IF KEY=52 THEN GOSUB 1210 ELSE 520
420 RETURN
430 PRINT :: INPUT "quanti records?":B :: GOSUB 440 :: GOTO 460
440 CALL CLEAR :: PRINT "PARAMETRI DI DEFAULT": "file #:2": "org:sequential": "type:internal": "len:fixed 140": "mode:input"
450 PRINT "lista stampa:alfan.4 input": : : RETURN
460 OPEN #2:"CS1",INTERNAL,INPUT,FIXED 140
470 FOR I=1 TO B :: ON ERROR 130 :: INPUT #2:A$,B$,C$,D$: : S$=A$&" "&B$&" "&C$&" "&D$: : IF S$="" THEN 500
480 F=3 :: GOSUB 780
490 IF S$="000" THEN 510
500 NEXT I
510 CLOSE #2 :: RETURN
520 PRINT :: LINPUT "inserisci il record " :S$ :: IF S$="" THEN RETURN
530 GOSUB 810
```

```

540 IF F=3 THEN GOSUB 780
550 IF F=1 THEN PRINT S$;"ATTENZ
IONE!e`gia` nel file!" ELSE 570
DOWNLOADED FROM WWW.TI99IUC.IT
560 CALL SOUND(200,500,0):: GOSU
B 930 :: RETURN
570 IF F=2 THEN PRINT "ATTENZION
E! il file e`gia` completo con
500 records!" ELSE RETURN
580 PRINT "|-ricerca in base al
numero 2-ricerca in base al nome
" :: GOSUB 1320 :: IF KEY=ASC
("2")THEN 620
590 INPUT "numero del record? ":
B :: IF N$(B)="" THEN PRINT "ATT
ENZIONE!questo record none`nel f
ile!" ELSE 610
600 CALL SOUND(200,500,0):: GOSU
B 930 :: F=5 :: RETURN
610 PRINT N$(B):: GOSUB 930 :: R
ETURN
620 GOSUB 1330 :: GOSUB 810 :: B
=K
630 IF F=3 THEN PRINT "ATTENZION
E! questo record` non e` nel fi
le!" ELSE 650
640 CALL SOUND(200,500,0):: GOSU
B 930 :: F=5 :: RETURN
650 PRINT : "numero del record:
";B:N$(B):: FOR W=1 TO 1000 :: N
EXT W
660 GOSUB 930 :: RETURN
670 GOSUB 850
680 GOSUB 930 :: RETURN
690 PRINT "|-ricerca in base al
numero 2-ricerca in base al nome
" :: GOSUB 1320 :: IF KEY=ASC("2
")THEN 710
700 INPUT "numero del record?":B
:: GOTO 740
710 GOSUB 1330 :: GOSUB 810 :: B
=K
720 IF F=3 THEN PRINT "ATTENZION
E! questo record` non e` nel fi
le!" ELSE 740
730 CALL SOUND(200,500,0):: GOSU
B 930 :: RETURN
740 GOSUB 880
750 PRINT "OK. il record e` canc
ellato"
760 GOSUB 930 :: RETURN
770 GOTO 950
780 REM AGGIUNGE UN RECORD
790 IF N=MA THEN F=2 :: RETURN
800 F=3 :: N=N+1 :: N$(N)=S$ ::
RETURN
810 REM RICERCA SEQUENZIALE
820 FOR K=1 TO N :: IF S$=SEG$(N
$(K),1,5)THEN F=1 :: RETURN

```

```

830 NEXT K
840 F=3 :: RETURN
850 REM DISPLAY IL FILE
860 FOR I=1 TO N :: IF N$(I)=""
THEN N$(I)=N$(I+1):: R=R+1
870 PRINT "rec. ";I;TAB(10);N$(I)
: : : NEXT I :: N=N-R :: R=0 ::
RETURN
880 REM CANCELLA UN RECORD
890 N=N-1 :: IF B=N+1 THEN RETUR
N
900 FOR I=B TO N :: N$(I)=N$(I+1
):: NEXT I :: RETURN
910 REM FINE
920 GOTO 950
930 FOR W=1 TO 800 :: NEXT W
940 RETURN
950 CALL CLEAR :: PRINT "vuoi re
gistrare il file? (S/N) " :: GOS
UB 1320 :: IF KEY=ASC("S")OR KEY
=ASC("s")THEN 960 ELSE 1020
960 CALL CLEAR :: PRINT "PARAMET
RI DI DEFAULT": : "org:sequent
ial": "file type:internal": "r
ecord type:fixed 140": "open mo
de:output"
970 PRINT "file #:1": "lista s
tampa:alfanumerica": : : OPEN
#: "CS1",INTERNAL,FIXED 140,OUT
PUT
980 PRINT #1:STR$(N):: CALL CLEA
R :: DISPLAY AT(2,3):"ATTESA" ::
DISPLAY AT(6,1)SIZE(4):"min." :
: DISPLAY AT(6,5)SIZE(4):INT(N*9
0/60)/10
990 FOR I=1 TO N :: ON ERROR 130
:: PRINT #1:N$(I):: DISPLAY AT(
6,5)SIZE(4):INT((N-I)*90/60)/10
:: IF N$(I)="000" THEN 1010
1000 NEXT I
1010 CLOSE #1
1020 END
1030 PRINT "quale record e` da m
odificare?" :: GOSUB 880
1040 IF F=5 THEN RETURN
1050 A$=SEG$(N$(B),1,28):: B$=SE
G$(N$(B),29,56):: C$=SEG$(N$(B),
57,84):: D$=SEG$(N$(B),85,112)::
E$=SEG$(N$(B),113,140):: PRINT
: "correggi riga per riga": :
1060 DISPLAY AT(24,1):A$ :: ACCE
PT AT(24,1)SIZE(-28):A$ :: DISPL
AY AT(24,1):B$ :: ACCEPT AT(24,1
)SIZE(-28):B$ :: DISPLAY AT(24,1
):C$ :: ACCEPT AT(24,1)SIZE(-28
):C$
1070 DISPLAY AT(24,1):D$ :: ACCE
PT AT(24,1)SIZE(-28):D$ :: DISPL
AY AT(24,1):E$ :: ACCEPT AT(24,1

```

```

)SIZE(-28):E$ :: N$(X)=A$&B$&C$&
D$&E$
1080 PRINT "OK il record corrett
o e`:" :N$(B):: FOR W=1 TO 1000
:: NEXT W :: RETURN
1090 REM ALPHA SORT
1100 FOR I=1 TO N-1 :: IF N$(I+1
)>=N$(I)THEN 1120
1110 A$=N$(I+1):: N$(I+1)=N$(I):
: N$(I)=A$ :: GOTO 1100
1120 NEXT I
1130 PRINT "OK.sono in ordine al
fabetico" :: FOR W=1 TO 1000 ::
NEXT W :: RETURN
1140 REM AGGIUNGE UN RECORD ALLA
POSIZIONE B
1150 PRINT :: INPUT "in quale po
sizione? ":B :: LINPUT "inserisc
i il record ":S$ :: GOSU
B 810
1160 IF F=3 THEN 1170 ELSE 1180.
DOWNLOADED FROM WWW.TI99IUC.IT
1170 FOR K=N TO B STEP -1 :: N$(
K+1)=N$(K):: NEXT K :: N$(B)=S$
:: N=N+1 :: RETURN
1180 IF F=1 THEN PRINT S$;"ATTEN
ZIONE!e` gia` nel file!" ELSE 12
00
1190 CALL SOUND(200,500,0):: GOS
UB 930 :: RETURN
1200 IF F=2 THEN PRINT "ATTENZIO
NE! il file e` gia` completo con
500 records!" ELSE RETURN
1210 PRINT : : :: PRINT "1-ricer
ca in base alla posiz2-ricerca i
n base al record": : :: GOSUB 13
20

```

```

1220 KEY=KEY-48 :: ON KEY GOSUB
1250,1280
1230 RETURN
1240 INPUT "quale posizione? ":B
:: GOSUB 440
1250 PRINT :: INPUT "in quale po
sizione? ":B
1260 OPEN #2:"CS1",INPUT ,FIXED
140,INTERNAL :: FOR I=1 TO (B-1)
:: ON ERROR 130 :: INPUT #2:S$ :
: NEXT I :: ON ERROR 130 :: INPU
T #2:A$,B$,C$,D$ :: N$(N+1)=A$&"
" &B$&" &C$&" &D$
1270 CLOSE #2 :: N=N+1 :: RETURN
1280 GOSUB 1330 :: PRINT :: INPU
T "quanti records probabilmentec
i sono nel file? ":B :: GOSUB 44
0
1290 OPEN #2:"CS1",INTERNAL,FIXE
D 140,INPUT :: FOR I=1 TO B :: O
N ERROR 130 :: INPUT #2:A$,B$,C$,
D$ :: E$=A$&" &B$&" &C$&" &D
$ :: IF SEG$(E$,1,5)=S$ THEN N$(
N+1)=E$ :: GOTO 1310
1300 NEXT I
1310 N=N+1 :: CLOSE #2 :: RETURN
1320 CALL KEY(0,KEY,STATUS):: IF
STATUS=0 THEN 1320 ELSE RETURN
1330 PRINT "batti l'inizio del r
ecord di5 caratteri ";TAB(18);"
" :: ACCEPT AT(23,13)SIZE(5):S$
:: RETURN

```

## Guida per l'input dei programmi sul TI 99/4A

A causa del sistema di codifica utilizzato, possono capitare in alcune linee delle improvvise "andate a capo" delle quali, naturalmente, non va tenuto conto in fase di digitazione.

Vi potrà capitare, a volte di trovare nei listati di programmi per TI 99/4a alcuni caratteri sottolineati. La sottolineatura rappresenta una particolare codifica dei tasti di controllo: dovrete quindi premere il tasto indicato dal carattere insieme al tasto CONTROL.

Ad esempio, se trovate A dovrete battere CONTROL+A; se trovate e—d, dovrete battere CONTROL+SPAZIO, e così via.



ritaglia  
e  
conserva  
ti  
attendono  
grosse  
sorprese!




 TI-99 ITALIAN USER CLUB

WWW.TI99IUC.IT

INFO@TI99IUC.IT

*Thanks to 99'er:*

*Antonio Stoppa*

*for this Scan.*

*Editing and Rework by:*

*TI99 Italian User Club in the year 2018*

*(info@ti99iuc.it)*

*Downloaded from [www.ti99iuc.it](http://www.ti99iuc.it)*

Per ricevere gli arretrati di **PAPER soft** compilare il seguente tagliando:

**J.soft** **Compilare ed inviare in busta chiusa a:**  
 via Rosellini 12 - 20142 Milano - Tel. 02/6888228-683797-6880841-2-3

Nome \_\_\_\_\_

Cognome \_\_\_\_\_

Indirizzo \_\_\_\_\_

CAP \_\_\_\_\_ Città \_\_\_\_\_ Prov. \_\_\_\_\_

Desidero ricevere i seguenti numeri arretrati di

**PAPER soft** \_\_\_\_\_

a L. 2.000 cad. per un totale di L. .... a mezzo

assegno allegato

contanti allegati

# PERSONAL COMPUTER: le prime, le migliori!


PERSONAL SOFTWARE

L'unica che presenta software  
per tutti i personal:

Commodore, Apple, Sinclair, T.I., HP,  
Sharp, Sega, Olivetti, ecc.



BIT: La più letta, la prima e più diffusa.  
TEST: Decision Mate V  
SUPERBIT - 64 pagine di programmi  
per i vostri personal computer.



Con tutta la competenza del  
GRUPPO EDITORIALE  
**JACKSON**